

ZX-81-Hardwaretip:

# Beim Gongschlag ist es 13 Uhr

Decoder für DCF-77-Zeitzeichen

Jetzt schlägt's 13: Eine handvoll Bauelemente und ein kurzes Programm reichen aus, um den ZX 81 zum Decoder für die Zeitzeichensignale des Senders DCF 77 zu machen. Holen auch Sie sich die „gesetzlich verordnete“ Uhrzeit ins Wohnzimmer.

Wenn zur abendlichen Stunde ein Gongschlag die „Tagesschau“ im Fernsehen einleitet, dann ist es 20 Uhr – und zwar haargenau. Die Tagesschau-Uhr ist nämlich per Funkweg mit dem Zeitzeichensender DCF 77 „verbunden“, der wiederum die in der Physikalisch Technischen Bundesanstalt, Braunschweig, gewonnene Normalzeit verbreitet. Und diese Normalzeit wird so genau eingehalten, daß übliche Quarzuhren im Vergleich dazu nur Sanduhren sind.

## Wochentag und Datum begleiten die Uhrzeit

Der Standort des Senders DCF 77 hat die Koordinaten 50°01' Nord und 9°00' Ost. Das ist etwa 24 km südöstlich von Frankfurt (Main) gelegen. Seit September 1970 arbeitet dieser Sender im Dauerbetrieb auf der Frequenz 77,5 kHz. Neben dieser hochgenauen Normalfrequenz (Fehler:  $10^{-15}$ ) liefert der Sender Sekundenmarken, und alle zwanzig Minuten wird sein Rufzeichen durch einen aufmodulierten 250-Hz-Ton in Morsetelegraphie übermittelt.

Seit Juni 1973 werden zusätzlich zu den Sekundenmarken binär codierte Informationen übertragen. Einer Sekundenmarke mit der Länge 0,1 s wird eine „0“ zugeordnet. Dauert die, durch eine Trägerabsenkung gebildete Marke 0,2 s,

dann stellt dies eine „1“ dar. Die codierten Informationen, die das Datum, die Uhrzeit, den Wochentag und weiteres umfassen, werden so bitseriell übertragen.

Um die Bits eindeutig auswerten zu können, ist noch eine Synchronisation

notwendig. Beim DCF-77-Signal erfolgt sie einfach durch Weglassen des 59sten Sekundenimpulses. Mit dem folgenden Impuls beginnt dann eine neue Minute. Für die Decodierung dieses komplexen Signals bietet sich ein Mikrocomputer geradezu an.


## Der EAR-Eingang empfängt die Zeitzeichen

Um die Zeitzeichen empfangen zu können, wird ein Empfänger für das Signal mit der Frequenz 77,5 kHz benötigt. Solche Empfänger sind als fertige Baugruppen erhältlich, aber auch der Eigenbau ist möglich [1], [2].

Der Empfänger sollte die Sekundenmarken in digitaler Form liefern. Dem einen logischen Pegel des Ausgangssignals ist dann die volle Trägeramplitude zugeordnet. Der andere logische Pegel entspricht der Trägerabsenkung mit 25 % der Maximalamplitude.

Um die Sekundenmarken ohne Eingriffe in das Innenleben des ZX 81 in diesen einspeisen zu können, wird noch ein kleines Interface benötigt. Es ist sehr einfach, weil der EAR-Eingang des

① **Interface:** Die Schaltung setzt das digitale Ausgangssignal eines DCF-77-Empfängers in ein NF-Signal um, so daß ein Einspeisen in die EAR-Buchse des ZX 81 möglich ist



② **Hex-Listing:** Das komplette Programm paßt noch in die 1-KByte-Version des ZX 81

```

16514: CD 08 41 30 FB 21 50 40 7E C6 01 27 FE 60 20 01
16530: AF CD 4B 41 3E 7F DB FE 1F D0 06 07 CD 01 41 CD
16546: 16 41 06 28 CD 01 41 06 06 38 DA C5 06 01 CD 01
16562: 41 59 C1 4B 10 F3 21 50 40 AF C6 59 CD 4B 41 21
16578: 45 40 CD 16 41 7E E6 3F 32 55 40 CD 16 41 2B 2B
16594: 7E E6 7F 32 51 40 23 7E E6 3F 32 52 40 23 23 7E
16610: E6 1F 32 54 40 06 05 CD 16 41 10 FB 7E 32 53 40
16626: 2B 7E E6 07 32 56 40 06 2C CD D1 41 C3 82 40 3A
16642: 34 40 8D B9 20 F9 3A 34 40 4F 06 32 DB FE 17 00
16658: D8 10 F9 C9 E5 C5 06 08 21 47 40 CB 1E 2B 10 FB
16674: C1 E1 C9 17 17 32 34 29 2E 32 2E 29 34 2B 37 38
16690: 26 38 34 1A CD 44 41 1A 1F 1F 1F 1F CD 44 41 2B
16706: 13 C9 E6 0F C6 1C 77 2B C9 F5 77 E5 2A 0C 40 11
16722: 16 00 19 D1 CD 35 41 F1 C0 06 05 CD 35 41 10 FB
16738: EB 7E 21 26 41 07 85 6F EB A8 ED A8 21 16 00 19
16754: 36 32 23 36 2A 23 3A 41 40 07 30 03 36 38 23 36
16770: 3F 23 36 00 23 36 00 07 30 02 36 A6 23 36 00 07
16786: 30 02 36 B7 23 36 00 C9
    
```

ZX 81 benutzt wird. Hier anliegende Wechselspannungssignale lassen sich mittels Software erfassen.


Das Interface (Bild 1) hat lediglich die Aufgabe, das digitale Ausgangssignal des DCF-77-Empfängers in ein Wechselspannungssignal umzuformen. Dazu wird ein Tonsignal mit einer Frequenz von etwa 1,5 bis 2 kHz erzeugt, das mit dem digitalen Zeitzeichensignal ein- und ausgeschaltet wird. Die logischen Pegel des steuernden Digitalsignals sollten dazu unter 1 V (L-Pegel) bzw. über 2,5 V (H-Pegel) liegen. DCF-77-Empfänger mit Ausgangsstufen in TTL- oder CMOS-Technik erfüllen diese Forderungen.

Die Diode D1 sorgt dafür, das auch das von Operationsverstärkern gelieferte  $\pm 15\text{-V}$ -Ausgangssignal benutzt werden kann. Das über den Schutzwiderstand R6 geführte NF-Ausgangssignal hat in jedem Fall einen für den EAR-Eingang ausreichenden Pegel.


Vor der Inbetriebnahme der Schaltung sollte mit einem Kopfhörer überprüft werden, ob die Sekundenmarken auch tatsächlich als kurze Piep-Töne am Ausgang des Interface anstehen. Wenn das Signal invertiert ist, d. h. die Sekundenmarken als Pausen im Tonsignal auftreten, dann muß dies im Programm berücksichtigt werden. Nach der Eingabe des Maschinencodes ist dazu der Befehl POKE 16657,63 einzugeben. Hardwareänderungen sind nicht erforderlich.

Mit 2 min „Fangzeit“ muß man rechnen

Das Maschinenprogramm wird in einer REM-Zeile abgelegt. Diese Zeile mit der Nummer 1 muß genau 280 Byte enthalten. Mit einem Hex-Lader (z. B. aus Heft 12/1983, Seite 76 oder Heft 20/1983, Seite 76) kann dann der Maschinencode (Bild 2) eingegeben werden. Danach darf man den Hex-Lader wieder löschen, da



③ **Struktogramm:** Bei genauerer Betrachtung kristallisiert sich das „Funktions-schaltbild“ des Programms heraus



④ **Speicherschema:** So ist die Zeitinformation nach 58 Sekundenmarken in einem 64-Bit-Schieberegister (bestehend aus acht Worten) vollständig geladen

- P1, P2, P3 = Parity-Bits (werden nicht ausgewertet)
- S = Startbit der Information (Zeit und Datum)
- SZ = Sommerzeit
- A = Ankündigung der Sommerzeit
- R = Reserveantenne
- M = Minutenanfang (Bit bei der Sekunde 0)
- X = z. Z. nicht codiert
- § = nicht benötigte Bits des Schieberegisters

er nicht mehr benötigt wird. Der Aufruf des Programms geschieht durch folgende Zeilen:

```
100 CLS
110 PRINT "=="WO=TT.MM.JJ=HH.MM:
    SS==ZEITSK=="
120 RAND USR 16514
```

Zeile 110 erzeugt in der ersten Bildschirmzeile eine Maske, in die vom Maschinenprogramm der Wochentag, das Datum, die Uhrzeit und weitere Daten eingetragen werden. Die Buchstaben haben lediglich die Funktion von Platzhaltern, und die Zeichen „=" sowie „:" sind nur Trennsymbole, die auch anders gewählt werden können. Das Zeichen „=" steht für SPACE. Da die genaue Anzahl der Space-Zeichen für die richtige Anzeige wichtig ist, wurde diese Darstellung in Zeile 110 benutzt.

Nachdem das Programm auf Kassette gesichert wurde, kann es gestartet werden (RUN 100). Der Sekundenzähler beginnt sofort mit dem Zählen der Zeitmarken. In der 59sten Sekunde wird der Zählerstand korrigiert und mit Beginn der neuen Minute werden die decodierten Daten angezeigt. Wenn von der ersten Minute mindestens 43 Sekundenmarken eingelesen wurden, dann sollte jetzt bereits die korrekte Information auf dem Schirm stehen. Andernfalls muß noch eine Minute gewartet werden.

Neben dem „Sommerzeitbit“, mit dem zwischen MEZ und MESZ umgeschaltet wird, werden noch zwei weitere Bits mit jeweils einem inversen Buchstaben zur Anzeige gebracht. Das „A“ steht dabei für das gesetzte Ankündigungsbit. Hierbei wird, vor dem Wechsel von MEZ auf MESZ und zurück, für die Dauer von einer Stunde mechanischen Uhren der Stundensprung mitgeteilt.

Das „R“ kennzeichnet den Betrieb des Senders über die Reserveantenne. Ein Umschalten der Antennen ist bei Wartungsarbeiten erforderlich und erfolgt phasensprungfrei. Je nach Empfangsort kann jedoch, durch die nun veränderte Laufzeit, dennoch ein Phasensprung von bis zu  $\pm 85$  Grad auftreten. Benutzer, die beim Auswerten der Signale höchste Genauigkeitsforderungen stellen, können sich so auf den aus der Umschaltung resultierenden Fehler einstellen.

Die Aussendungen des DCF 77 gelten nach Aussage der Physikalisch-Technischen Bundesanstalt und des Bundesministers für das Post- und Fernmeldewesen als langfristig gesichert [3]. Seit 1978 ist der Empfang des Zeitzeichensenders lizenz- und gebührenfrei [4].

## Ein Schieberegister schluckt die Zeitdaten

Zur Beschreibung des Programms dient ein Struktogramm (Bild 3), also ein grafischer Funktionsplan. Das darin enthaltene Unterprogramm „Display“ meldet den Stand des Sekundenzählers auf dem Bildschirm. Bei der Sekunde 0 zeigt es zusätzlich die in der letzten Minute empfangene und decodierte Zeitinformation an.


Ein anderes Unterprogramm bewirkt Wartezeiten. Diese Zeiten sind immer Vielfache von 20 ms, da dieses Zeitraster vom ZX 81 vorgegeben wird. Die Systemvariable FRAMES, die die am Bildschirm ausgegebene Halbbilder zählt, wird alle 20 ms dekrementiert und vom Maschinenprogramm überwacht.

Wenn nach 58 Sekunden alle Bits der kompletten Zeitinformation im Schieberegister stehen, dann kann der Inhalt des

Schieberegisters decodiert werden. Bild 4 zeigt die dann im Schieberegister stehenden Daten, Bild 5 veranschaulicht das Prinzip der Decodierung. Tiefere Einblicke in die Funktion des Programms gewährt Kennern der Maschinensprache das knapp kommentierte Assembler-Listing, das vom Franzis-Software-Service gegen DM 4,20 in Briefmarken angefordert werden kann (Stichwort: DCF-77-Decoder mit ZX 81; FUNKSCHAU 22/1984). Joachim Börke

### Literatur

- [1] Hetzel, P.; Lohbeck, L.: Datums- und Zeitangabe drahtlos empfangen. FUNKSCHAU 19/1974.
- [2] Gößler, R.: Lautloses Ticken. Ein Miniatur-Zeitzeichen-Empfänger. ELO 9/1982.
- [3] Becker, G.: Mitteilungen über den Sender DCF 77. PTB-Mitteilungen 87, Heft 6/1977, Seite 542.
- [4] Amtsblatt des BPM Nr. 34/1978.
- [5] Gesetz über die Zeitbestimmung (ZeitG). Bundesgesetzblatt (1978), Teil 1, Seite 1110.
- [6] Becker, G.; Hetzel, P.: Codierung von Sommerzeit und anderen Informationen beim Sender DCF 77. PTB-Mitteilungen 87, Heft 2/1977, Seite 110-112.



### Darstellung der decodierten Daten

Bit	2 <sup>7</sup>	2 <sup>6</sup>	2 <sup>5</sup>	2 <sup>4</sup>	2 <sup>3</sup>	2 <sup>2</sup>	2 <sup>1</sup>	2 <sup>0</sup>
	8	4	2	1	8	4	2	1
	Zehner				Einer			

Adresse	Inhalt	Wertebereich
4050H	Sekunden	00...59
4051H	Minuten	00...59
4052H	Stunden	00...23
4053H	Jahr	00...99
4054H	Monat	01...12
4055H	Tag	01...31
4056H	Wochentag	01...07 (01 $\triangleq$ Montag)

⑤ **Decodierschema:** Das Struktogramm (oben) beschreibt die Decodierung von Uhrzeit und Datum. Das Ergebnis wird in sieben Speicherzellen (unten) abgelegt

```

1      ; DEKODIERUNG DES DCF-77 SIGNALS MIT
2      ; EINEM MICROCOMPUTER ZX81
3      ;
4      ;
5      ; JOACHIM BOERKE      10.05.1984
6      ;
7      ;
8      ; ZX81 SPEZIFISCHE VEREINBARUNGEN
00FE   9 EAR      EQU 00FEH
400C  10 DFILE   EQU 400CH
4034  11 FRAMES  EQU 4034H
12      ; VEREINBARUNG DER VARIABLEN
4040  13 SR      EQU 4040H
4050  14 SEKUND  EQU 4050H
4051  15 MINUTE  EQU SEKUND+1
4052  16 STUNDE  EQU SEKUND+2
4053  17 JAHR    EQU SEKUND+3
4054  18 MONAT   EQU SEKUND+4
4055  19 TAG     EQU SEKUND+5
4056  20 WOTAG   EQU SEKUND+6
21      ;
22      ; PROGRAMMAUFRUF MIT USR(16514)
4082  23      ORG 16514
4082  24 START:  CALL SIGNAL
4085  25      JR NC,START ; AUF SEKUNDENBEGINN WARTEN
4087  26 FLANKE: LD HL,SEKUND
408A  27      LD A,(HL)
408B  28      ADD A,1      ; SEKUNDEN := SEKUNDEN + 1
408D  29      DAA
408E  30      CP 60H      ; SEKUNDEN = 60 ?
4090  31      JR NZ,NE60
4092  32      XOR A
4093  33 NE60:  CALL DISPLAY
4096  34      LD A,7FH      ; BREAK-TASTE
4098  35      IN A,EAR      ; ABFRAGEN
409A  36      RRA
409B  37      RET NC      ; ZURUECK ZUM BASIC
409C  38      LD B,7
409E  39      CALL PAUSE   ; 0,14 S
40A1  40      CALL SHIFT
40A4  41      LD B,40
40A6  42      CALL PAUSE   ; 0,80 S
40A9  43      LD B,6
40AB  44 LOOP1: JR C,FLANKE ; ZURUECK, WENN SIGNAL VORHANDEN
40AD  45      PUSH BC
40AE  46      LD B,1
40B0  47      CALL PAUSE   ; 0,02 S
40B3  48      LD E,C
40B4  49      POP BC
40B5  50      LD C,E
40B6  51      DJNZ LOOP1  ; MAX 0,12 S AUF SIGNAL WARTEN
40B8  52      LD HL,SEKUND ; SEKUNDENIMPULS FEHLT
40BB  53      XOR A
40BC  54      ADD A,59H    ; SEKUNDEN := 59
40BE  55      CALL DISPLAY
56      ;
40C1  57      LD HL,SR+5   ; ZEIT UND DATUM DEKODIEREN
40C4  58      CALL SHIFT
40C7  59      LD A,(HL)   ; SR+5
40C8  60      AND 3FH

```

LOC	OBJ	CODE	STMT	SOURCE	STATEMENT
40CA	325540		61		LD (TAG),A
40CD	CD1641		62		CALL SHIFT
40D0	2B		63		DEC HL
40D1	2B		64		DEC HL
40D2	7E		65		LD A,(HL) ; SR+3
40D3	E67F		66		AND 7FH
40D5	325140		67		LD (MINUTE),A
40D8	23		68		INC HL
40D9	7E		69		LD A,(HL) ; SR+4
40DA	E63F		70		AND 3FH
40DC	325240		71		LD (STUNDE),A
40DF	23		72		INC HL
40E0	23		73		INC HL
40E1	7E		74		LD A,(HL) ; SR+6
40E2	E61F		75		AND 1FH
40E4	325440		76		LD(MONAT),A
40E7	0605		77		LD B,05
40E9	CD1641		78	LOOP2:	CALL SHIFT
40EC	10FB		79		DJNZ LOOP2
40EE	7E		80		LD A,(HL) ; SR+6
40EF	325340		81		LD (JAHR),A
40F2	2B		82		DEC HL
40F3	7E		83		LD A,(HL) ; SR+5
40F4	E607		84		AND 07H
40F6	325640		85		LD (WOTAG),A ; DEKODIERUNG BEENDET
			86		;
40F9	062C		87		LD B,44
40FB	CD0141		88		CALL PAUSE ; 0,88 S
40FE	C38240		89		JP START
			90		;
4101	3A3440		91	PAUSE:	LD A,(FRAMES) ; B * 0,02 S PAUSE
4104	80		92		ADD A,B
4105	B9		93		CP C
4106	20F9		94		JR NZ PAUSE
			95		;
4108	3A3440		96	SIGNAL:	LD A,(FRAMES) ; ANFANGSZEIT DER NAECHSTEN
410B	4F		97		LD C,A ; PAUSE IN REGISTER C ABLEGEN
410C	0632		98		LD B,50 ; OHRHOEREREINGANG ABFRAGEN
410E	DBFE		99	LOOP3:	IN A,EAR
4110	17		100		RLA
4111	00		101		NOP ; EVTL. DURCH CCF ERSETZEN
4112	D8		102		RET C ; SIGNAL VORHANDEN
4113	10F9		103		DJNZ LOOP3
4115	C9		104		RET ; KEIN SIGNAL VORHANDEN
			105		;
4116	E5		106	SHIFT:	PUSH HL ; CY IN SR SCHIEBEN
4117	C5		107		PUSH BC
4118	0608		108		LD B,08H ; SR BESTEHT AUS 8 BYTE
411A	214740		109		LD HL,SR+7
411D	CB1E		110	LOOP4:	RR (HL)
411F	2B		111		DEC HL
4120	10FB		112		DJNZ LOOP4
4122	C1		113		POP BC
4123	E1		114		POP HL
4124	C9		115		RET
			116		;
4125	1717		117	TAB:	DEFW 1717H ; **
4127	3234		118		DEFW 3432H ; MO
4129	292E		119		DEFW 2E29H ; DI
412B	322E		120		DEFW 2E32H ; MI

```

412D 2934      121      DEFW 3429H ; DO
412F 2B37      122      DEFW 372BH ; FR
4131 3826      123      DEFW 2638H ; SA
4133 3834      124      DEFW 3438H ; SO
                125      ;
4135 1A        126 ANZ:   LD A,(DE)  ; ZWEI ZIFFERN LADEN
4136 CD4441    127      CALL UMWA  ; NIEDERWERTIGE ZIFFER ANZEIGEN
4139 1A        128      LD A,(DE)
413A 1F        129      RRA
413B 1F        130      RRA
413C 1F        131      RRA
413D 1F        132      RRA
413E CD4441    133      CALL UMWA  ; HOEHERWERTIGE ZIFFER ANZEIGEN
4141 2B        134      DEC HL    ; ZWISCHENRAUM
4142 13        135      INC DE   ; ZEIGER AUF NAECHSTES BYTE
4143 C9        136      RET
                137      ;
4144 E60F      138 UMWA:   AND OFH   ; HALBBYTE
4146 C61C      139      ADD A,1CH ; IN ZX-CODE UMWANDELN
4148 77        140      LD (HL),A ; UND ANZEIGEN
4149 2B        141      DEC HL   ; NAECHSTE SPALTE
414A C9        142      RET
                143      ;
414B F5        144 DISPLAY: PUSH AF
414C 77        145      LD (HL),A ; SEKUND VERAENDERN
414D E5        146      PUSH HL
414E 2A0C40    147      LD HL,(DFILE)
4151 111600    148      LD DE,0016H
4154 19        149      ADD HL,DE ; SPALTE FUER SEKUNDEN-EINER
4155 D1        150      POP DE
4156 CD3541    151      CALL ANZ  ; SEKUNDEN ANZEIGEN
4159 F1        152      POP AF
415A C0        153      RET NZ   ; NUR SEKUNDEN VERAENDERN
                154      ;
415B 0605      155      LD B,05H  ; ZEIT UND DATUM ANZEIGEN
415D CD3541    156 LOOP5:  CALL ANZ
4160 10FB      157      DJNZ LOOP5
4162 EB        158      EX DE,HL ; WOCHENTAGSKUERZEL ANZEIGEN
4163 7E        159      LD A,(HL) ; WOTAG
4164 212641    160      LD HL,TAB+1
4167 07        161      RLCA    ; A:=A*2
4168 85        162      ADD A,L
4169 6F        163      LD L,A   ; TABELLENPLATZ BESTIMMEN
416A EDA8      164      LDD    ; 2. BUCHSTABE
416C EDA8      165      LDD    ; 1. BUCHSTABE
416E 211600    166      LD HL,0016H ; ZEITSKALA ME(S)Z ANZEIGEN
4171 19        167      ADD HL,DE ; SPALTE VON "M"
4172 3632      168      LD (HL),32H ; "M"
4174 23        169      INC HL
4175 362A      170      LD (HL),2AH ; "E"
4177 23        171      INC HL
4178 3A4140    172      LD A,(SR+1)
417B 07        173      RLCA    ; SOMMERZEITBIT NACH CY
417C 3003      174      JR NC,MEZ
417E 3638      175      LD (HL),38H ; "S"
4180 23        176      INC HL
4181 363F      177 MEZ:  LD (HL),3FH ; "Z"
4183 23        178      INC HL
4184 3600      179      LD (HL),00H ; " "
4186 23        180      INC HL

```

LOC	OBJ	CODE	STMT	SOURCE	STATEMENT
4187	3600		181		LD (HL),00H ;" "
4189	07		182		RLCA ; ANKUENDIGUNGSBIT NACH CY
418A	3002		183		JR NC,NOANK
418C	36A6		184		LD (HL),0A6H ;"A"
418E	23		185	NOANK:	INC HL
418F	3600		186		LD (HL),00H ;" "
4191	07		187		RLCA ; RESERVEANTENNENBIT NACH CY
4192	3002		188		JR NC,NORES
4194	36B7		189		LD (HL),0B7H ;"R"
4196	23		190	NORES:	INC HL
4197	3600		191		LD (HL),00H ;" "
4199	C9		192		RET
			193		;
419A	0000		194		END

0 ASSEMBLY ERRORS

SYMBOL	VAL.	LINE							
ANZ	4135	126	151	156					
DFILE	400C	10	147						
DISPLA	414B	33	55	144					
EAR	00FE	9	35	99					
FLANKE	4087	26	44						
FRAMES	4034	11	91	96					
JAHR	4053	17	81						
LOOP1	40AB	44	51						
LOOP2	40E9	78	79						
LOOP3	410E	99	103						
LOOP4	411D	110	112						
LOOP5	415D	156	157						
MEZ	4181	174	177						
MINUTE	4051	15	67						
MONAT	4054	18	76						
NE60	4093	31	33						
NOANK	418E	183	185						
NORES	4196	188	190						
PAUSE	4101	39	42	47	88	91	94		
SEKUND	4050	14	15	16	17	18	19	20	26
		52							
SHIFT	4116	40	58	62	78	106			
SIGNAL	4108	24	96						
SR	4040	13	57	109	172				
START	4082	24	25	89					
STUNDE	4052	16	71						
TAB	4125	117	160						
TAG	4055	19	61						
UMWA	4144	127	133	138					
WOTAG	4056	20	85						